

Erraten Sie die Koeffizienten einer Sinusfunktion (TI-82 STATS)

(Aus der [Bedienungsanleitung](#) – Anwendungsbeispiele 17–13)

Dieses Programm zeichnet die Funktion $A \sin(BX)$ mit ganzzahligen Zufallskoeffizienten zwischen 1 und 10 für A und B. Versuchen Sie die Koeffizienten zu erraten und lassen Sie Ihre Schätzungen als $C \sin(DX)$ graphisch darstellen. Das Programm wird solange fortgesetzt, bis Ihre Schätzung richtig ist.

Die Eingabe starten Sie mit [PRGM][→][→] NEW 1:Create New [ENTER] „RATEN“
spezielle Symbole und Befehle findet man allgemein unter [2nd] [CATALOG über 0] und dem Anfangsbuchstaben. Schließen Sie jede Programmzeile mit [ENTER] ab!
„→“ bedeutet [STO>] „Speichern“

PROGRAMM: RATEN

:PlotsOff	Abschalten der PLOTs
:Func	Funktionsdarstellung einschalten
:FnOff	bisherige Funktionen ausschalten
:Radian	auf Radian umstellen (von Degree)
:ClrHome	löscht den Hauptbildschirm
:"Asin(BX)"→Y1	Definition der Zufalls-Gleichung
:"Csin(DX)"→Y2	Definition der geratenen Gleichung
:GraphStyle(1,1)	Zufalls-Funktion mit „Linie“ zeichnen
:GraphStyle(2,5)	geratene Funktion mit „Verlauf mit Kugel“ zeichnen
:FnOff 2	geratene Funktion zuerst abschalten
:randInt(1,10)→A	Zufallszahl für A eingeben
:randInt(1,10)→B	Zufallszahl für B eingeben
:0→C	C Null setzen
:0→D	D Null setzen
:-2π→Xmin	WINDOW-Variable Xmin setzen [VARS] [1:WINDOW] [1:Xmin]
:2π→Xmax	WINDOW-Variable Xmax setzen [VARS] [1:WINDOW] [2:Xmax]
:π/2→Xscl	WINDOW-Variable Xscl setzen [VARS] [1:WINDOW] [2:Xscl]
:-10→Ymin	WINDOW-Variable Ymin setzen [VARS] [1:WINDOW] [4:Ymin]
:10→Ymax	WINDOW-Variable Ymax setzen [VARS] [1:WINDOW] [5:Ymax]
:1→Yscl	WINDOW-Variable Yscl setzen [VARS] [1:WINDOW] [6:Yscl]
:DispGraph	Graph anzeigen
:Pause	Warten auf „ENTER“
:FnOn 2	geratene Funktion einschalten (mit 0,0)
:Lbl Z	Beginn der Schleife (LABEL Z)
:Prompt C,D	Eingabe-Aufforderung für die Schätzung
:DispGraph	ingegebenen Graph anzeigen
:Pause	Warten auf „ENTER“ für die Bewertung
:If C=A	wenn C richtig ist
:Text(1,1,"C IST OK")	
:If C≠A	wenn C falsch ist
:Text(1,1,"C IST FALSCH")	
:If D=B	wenn D richtig ist
:Text(1,50,"D IST OK")	
:If D≠B	wenn D falsch ist
:Text(1,50,"D IST FALSCH")	
:DispGraph	Anzeige der Beurteilung
:Pause	Warten auf „ENTER“
:If C=A and D=B	wenn C und D richtig geraten sind
:Then	dann
:Degree	stellt auf Degree (für die Trigonometrie) um
:DiagnosticOn	schaltet den Diagnosemodus bei (Linearen) Regressionen ein
:ClrHome	löscht den Hauptbildschirm
:Stop	und stoppt
:End	Ende der THEN-Bedingung
:Goto Z	wenn noch nicht alles richtig ist, zurück zu LABEL Z